3.4: Applied Optimization Worksheet

1. A Norman window has the shape of a rectangle with a semicircle on top:



A Norman window needs to have a perimeter of 20 feet. Find its dimensions if it should let it in a maximum amount of light (i.e enclose maximal area).

```
In[]:= (* We constrain the perimeter, which allows us to write h as a function of r: *)
       perimeter = 20;

In[]:= (* The area is a function of height h, and radius r: *)
       A[r_, h_] := 2 r h + Pi r^2 / 2 (* a rectangle and half a circle *)

In[]:= (* We can write h as a function of r: *)
       h[r_] := (perimeter - Pi r - 2 r) / 2

In[]:= (* Then we differentiate A(r), and find its critical numbers: *)
       soln = D[A[r, h[r]], r]
       rval = r /. Solve[soln == 0, r][[1]]
       N[rval]
```

Out[]= $20 - 2\,r + (-2 - \pi)\,r$

Out[]= $\frac{20}{4 + \pi}$

Out[]= 2.8005

```
In[]:= (* Let's compute the area of the critical point: *)
       N[A[rval, h[rval]]]
```

Out[]= 28.005

In[ ]:=
```
(* Have a look at the area function over the domain of r: *)
Show[
 Plot[A[r, h[r]], {r, 0, 20 / (Pi + 2)}],
 ListPlot[{{rval, A[rval, h[rval]]}}]
]
```

Out[ ]=



AC 3.4.6. A rectangular box with a square bottom and closed top is to be made from two materials. The material for the side costs \$1.50 per square foot and the material for the top and bottom costs \$3.00 per square foot. If you are willing to spend \$15 on the box, what is the largest volume it can contain? Justify your answer completely using calculus.

In[ ]:=
```
sideCost = 1.5 ;
endCost = 3;
totalCost = 15;
```

In[ ]:=
```
(* The volume is a function of two variables, as is the cost: *)
V[s_, h_] := s^2 h
cost[s_, h_] :=
 (* top and bottom *) 2 s^2 * endCost +
  (* four sides *) 4 s h * sideCost
```

In[ ]:=
```
(* However we can solve to get the height, say, in terms of the square side length s: *)
height = h /. Solve[cost[s, h] == totalCost, h][[1]];
h[s_] = height
```

Out[ ]= $\frac{0.166667\,(15. - 6.\,s^2)}{s}$

In[ ]:=
```
(* Now we can figure out the limits on s, since h must be positive: *)
slimits = Solve[height == 0, s]
slimit = s /. slimits[[2]]
(* Check against what I got by hand: *)
N[Sqrt[15 / (2 endCost)]]
```

Out[ ]= {{s → -1.58114}, {s → 1.58114}}

Out[ ]= 1.58114

Out[ ]= 1.58114

In[ ]:=
```
(* Find the derivative of V, and set it equal to 0 for critical numbers: *)
soln = D[V[s, h[s]], s]
rval = s /. Solve[soln == 0, s][[2]]
N[rval]
(* Check against what I got by hand: *)
Sqrt[15 / 18] * 1.0
```

Out[ ]= $-2.\ s^2 + 0.166667\ (15. - 6.\ s^2)$

Out[ ]= 0.912871

Out[ ]= 0.912871

Out[ ]= 0.912871

In[ ]:=
```
(* Let's take a look, and get a summary of the solution: *)
Show[
 Plot[V[s, h[s]], {s, 0, slimit}],
 ListPlot[{{rval, V[rval, h[rval]]}}]
]
rval
h[rval]
N[V[rval, h[rval]]]
N[cost[rval, h[rval]]]
```

Out[ ]=


Out[ ]= 0.912871

Out[ ]= 1.82574

Out[ ]= 1.52145

Out[ ]= 15.

AC 3.4.9. A company is designing propane tanks that are cylindrical with hemispherical ends. Assume that the company wants tanks that will hold 1000 cubic feet of gas, and that the ends are more expensive to make, costing \$5 per square foot, while the cylindrical barrel between the ends costs \$2 per square foot. Use calculus to determine the minimum cost to construct such a tank.

(Surface area of a sphere: $4\pi r^2$)
(Volume of a sphere: $\frac{4}{3}\pi r^3$)

In[ ]:=
```
(* Here are the costs: *)
sideCost = 2;
endCost = 5;
```

In[ ]:=
```
(* The volume is the sum of a sphere, and the cylindrical part:  *)
V[r_, L_] := 4 / 3 Pi r^3 + Pi r^2 L
```

In[ ]:=
```
(* We know that the volume is constrained to
 be 1000 cubic feet: solve for L as a function of r: *)
soln = Solve[V[r, L] == 1000, L];
lof[r_] = L /. soln[[1]]
```

Out[ ]= $-\frac{4\left(-750+\pi r^3\right)}{3\pi r^2}$

In[ ]:=
```
(* The cost is a function of r and L (but we know L in terms of r): *)
cost[r_, lval_] := 4 Pi r^2 endCost + 2 Pi r lof[r] * sideCost
deriv[r_] = D[cost[r, lof[r]], r]
Plot[{cost[r, lof[r]], deriv[r]}, {r, 0, 5}]
```

Out[ ]= $24\pi r+\frac{16\left(-750+\pi r^3\right)}{3 r^2}$

Out[ ]=


In[ ]:=
```
(* Solve for the critical numbers: *)
soln = D[cost[r, lof[r]], r]
rval = r /. Solve[soln == 0, r][[2]]
rval = 5 * E^(1 / 3 * Log[12 / (11 * Pi)])
```

Out[ ]= $24\pi r+\frac{16\left(-750+\pi r^3\right)}{3 r^2}$

Out[ ]= $5\ (-2)^{2/3}\left(\frac{3}{11\pi}\right)^{1/3}$

Out[ ]= $5\times 2^{2/3}\left(\frac{3}{11\pi}\right)^{1/3}$

In[ ]:=
```
(* I got a solution by hand, and want to compare it to Mathematica's: *)
root = N[(16 * 750 / 88 / Pi) ^ (1 / 3)]
lof[root]
cost[root, lof[root]]
V[root, lof[root]] (* Check that the volume is still 1000 *)
```

Out[ ]= 3.51439

Out[ ]= 21.0863

Out[ ]= 1707.27

Out[ ]= 1000.

In[ ]:=
```
(* Checks out! *)
N[rval]
N[lof[rval]]
N[cost[rval, lof[rval]]]
Show[
 Plot[{cost[r, lof[r]], deriv[r]}, {r, 0, 7}],
 ListPlot[{{rval, cost[rval, lof[rval]]}}], PlotRange → {0, 4000}
]
```

Out[ ]= 3.51439

Out[ ]= 21.0863

Out[ ]= 1707.27

Out[ ]=